Problem:
Score : 200 points

Problem Statement

Given is a sequence of N integers A_1, \ldots, A_N.

Find the (multiplicative) inverse of the sum of the inverses of these numbers, \frac{1}{\frac{1}{A_1} + \ldots + \frac{1}{A_N}}.

Constraints

1 \leq N \leq 100

1 \leq A_i \leq 1000

Input

Input is given from Standard Input in the following format:

N
A_1 A_2 \ldots A_N

Output

Print a decimal number (or an integer) representing the value of \frac{1}{\frac{1}{A_1} + \ldots + \frac{1}{A_N}}.

Your output will be judged correct when its absolute or relative error from the judge's output is at most 10^{-5}.

Sample Input 1

2
10 30

Sample Output 1

7.5

\frac{1}{\frac{1}{10} + \frac{1}{30}} = \frac{1}{\frac{4}{30}} = \frac{30}{4} = 7.5.

Printing 7.50001, 7.49999, and so on will also be accepted.

Sample Input 2

3
200 200 200

Sample Output 2

66.66666666666667

\frac{1}{\frac{1}{200} + \frac{1}{200} + \frac{1}{200}} = \frac{1}{\frac{3}{200}} = \frac{200}{3} = 66.6666....

Printing 6.66666e+1 and so on will also be accepted.

Sample Input 3

1
1000

Sample Output 3

1000

\frac{1}{\frac{1}{1000}} = 1000.

Printing +1000.0 and so on will also be accepted.

Solution in Fortran (gfortran v4.8.4) (judge verdict: Accepted)

module ABC138

  ! <module>s to import
  use, intrinsic :: iso_fortran_env

  ! require all variables to be explicitly declared
  implicit none

  ! accessibility of <subroutine>s and <function>s in this <module>
  public  :: task_B
  private :: calc_solution

  ! contained <subroutine>s and <function>s are below
  contains

  subroutine task_B

    ! variables for this <subroutine>
    integer :: len_data

    ! arrays for this <subroutine>
    real(REAL64), dimension(:), allocatable :: val_data

    ! STEP.01
    ! read out the given data (length of the given data)
    read(unit=INPUT_UNIT, fmt=*) len_data

    ! STEP.02
    ! allocate the array to store the given data
    allocate( val_data(1:len_data) )

    ! STEP.03
    ! read out given data (value)
    read(unit=INPUT_UNIT, fmt=*) val_data(:)

    ! STEP.04
    ! calculate & output the answer of this task
    write(unit=OUTPUT_UNIT, fmt='(ES23.15e3)', advance='yes') calc_solution( val_data(:) )

    ! STEP.05
    ! deallocate the array to store the given data
    deallocate( val_data )

    ! STEP.END
    return

  end subroutine task_B

  pure function calc_solution (array) result(solution)

    ! arguments for this <function>
    real(REAL64), dimension(:), intent(in) :: array

    ! return value of this <function>
    real(REAL64) :: solution

    ! support variables for this <function>
    integer :: itr

    solution = 0.0_REAL64

    do itr = 1, size(array(:), dim=1), 1
      solution = solution + 1.0_REAL64 / array(itr)
    end do

    solution = 1.0_REAL64 / solution

  end function calc_solution

end module ABC138


program main

  ! <module>s to import
  use, non_intrinsic :: ABC138

  ! require all variables to be explicitly declared
  implicit none

  call task_B

end program main